PCF.index

November 29, 2025

Contents

1 PCF.Constants

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}

module PCF.Constants where

open import Data.Bool.Base
  using (Bool; true; false; if_then_else_)
open import Agda.Builtin.Nat
  using (Nat; _+_; _-_; _==_)

open import PCF.Domain-Notation
  using (⟪_⟫; η; _SHARP; fix; ⊥; _⟶_,_)
open import PCF.Types
  using (Types; o; ι; _⇒_; σ; 𝒟)

-- Syntax

data ℒ : Types → Set where
  tt  : ℒ o
  ff  : ℒ o
  ⊃ᵢ  : ℒ (o ⇒ ι ⇒ ι ⇒ ι)
  ⊃ₒ  : ℒ (o ⇒ o ⇒ o ⇒ o)
  Y   : {σ : Types} → ℒ ((σ ⇒ σ) ⇒ σ)
  k   : (n : Nat) → ℒ ι
  +1′ : ℒ (ι ⇒ ι)
  -1′ : ℒ (ι ⇒ ι)
  Z   : ℒ (ι ⇒ o)

variable c : ℒ σ

-- Semantics

𝒜⟦_⟧ : ℒ σ → ⟪ 𝒟 σ ⟫

𝒜⟦ tt  ⟧ = η true
𝒜⟦ ff  ⟧ = η false
𝒜⟦ ⊃ᵢ  ⟧ = _⟶_,_
𝒜⟦ ⊃ₒ  ⟧ = _⟶_,_
𝒜⟦ Y   ⟧ = fix
𝒜⟦ k n ⟧ = η n
𝒜⟦ +1′ ⟧ = (λ n → η (n + 1)) SHARP
𝒜⟦ -1′ ⟧ = (λ n → if n == 0 then ⊥ else η (n - 1)) SHARP
𝒜⟦ Z   ⟧ = (λ n → η (n == 0)) SHARP
```

2 PCF.Domain-Notation

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}
open import Agda.Builtin.Equality
open import Agda.Builtin.Equality.Rewrite

module PCF.Domain-Notation where

open import Relation.Binary.PropositionalEquality.Core
  using ( _≡_ ) public

------------------------------------------------------------------------
-- Domains

postulate
  Domain : Set₁
  ⟪_⟫    : Domain → Set

variable
  D E : Domain
  P : Set
  d₁ d₂ : Set

postulate
  ⊥ : ⟪ D ⟫ -- bottom element

------------------------------------------------------------------------
-- Function domains

postulate
  _→ᶜ_   : Domain → Domain → Domain -- assume continuous
  _→ˢ_   : Set → Domain → Domain    -- always continuous
  dom-cts : ⟪ D →ᶜ E ⟫ ≡ (⟪ D ⟫ → ⟪ E ⟫)
  set-cts : ⟪ P →ˢ E ⟫ ≡ (P → ⟪ E ⟫)

{-# REWRITE dom-cts set-cts #-}

postulate
  fix : ⟪ (D →ᶜ D) →ᶜ D ⟫ -- fixed point of endofunction

  -- Properties
  fix-fix : (f : ⟪ D →ᶜ D ⟫) → fix f ≡ f (fix f)

-- Flat domains

postulate
  _+⊥      : Set → Domain                      -- lifted set
  η        : ⟪ P →ˢ P +⊥ ⟫                     -- inclusion
  _SHARP   : ⟪ (P →ˢ D) →ᶜ P +⊥ →ᶜ D ⟫         -- Kleisli extension

  -- Properties
  elim-SHARP-η : (f : ⟪ P →ˢ D ⟫) (p : P) → (f SHARP) (η p) ≡ f p
  elim-SHARP-⊥ : (f : ⟪ P →ˢ D ⟫) →         (f SHARP) ⊥     ≡ ⊥
```

```
-- McCarthy conditional

-- t ⟶ d₁ , d₂ : ⟪ D ⟫  (t : Bool +⊥ ; d₁, d₂ : ⟪ D ⟫)

open import Data.Bool.Base
  using (Bool; true; false; if_then_else_) public

postulate
  _⟶_,_ : ⟪ Bool +⊥ →ᶜ D →ᶜ D →ᶜ D ⟫ -- McCarthy conditional

  -- Properties
  true-cond   : {d₁ d₂ : ⟪ D ⟫} → (η true ⟶ d₁ , d₂)        ≡ d₁
  false-cond  : {d₁ d₂ : ⟪ D ⟫} → (η false ⟶ d₁ , d₂) ≡ d₂
  bottom-cond : {d₁ d₂ : ⟪ D ⟫} → (⊥ ⟶ d₁ , d₂)             ≡ ⊥

infixr 0   _→ᶜ_
infixr 0   _→ˢ_
infix 10        _+⊥
infixr 20   _⟶_,_
```

3 PCF.Environments

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}

module PCF.Environments where

open import Data.Bool.Base
  using (Bool; if_then_else_)
open import Data.Maybe.Base
  using (Maybe; just; nothing)
open import Agda.Builtin.Nat
  using (Nat; _==_)
open import Relation.Binary.PropositionalEquality.Core
  using (_≡_; refl; trans; cong)

open import PCF.Domain-Notation
  using (⟪_⟫; ⊥)
open import PCF.Types
  using (Types; ι; o; _⇒_; 𝒟)
open import PCF.Variables
  using (𝒱; var; Env)

-- ρ⊥ is the initial environment

ρ⊥ : Env
ρ⊥ α = ⊥

-- (ρ [ x / α ]) α′ = x when α and α′ are identical, otherwise ρ α′

_[_/_] : {σ : Types} → Env → ⟪ 𝒟 σ ⟫ → 𝒱 σ → Env
ρ [ x / α ] = λ α′ → h ρ x α α′ (α ==V α′) where

  h : {σ τ : Types} → Env → ⟪ 𝒟 σ ⟫ → 𝒱 σ → 𝒱 τ → Maybe (σ ≡ τ) → ⟪ 𝒟 τ ⟫
  h ρ x α α′ (just refl) = x
  h ρ x α α′ nothing     = ρ α′

  _==T_ : (σ τ : Types) → Maybe (σ ≡ τ)
  (σ ⇒ τ) ==T (σ′ ⇒ τ′) = f (σ ==T σ′) (τ ==T τ′) where
          f : Maybe (σ ≡ σ′) → Maybe (τ ≡ τ′) → Maybe ((σ ⇒ τ) ≡ (σ′ ⇒ τ′))
          f = λ { (just p) (just q) → just (trans (cong (_⇒ τ) p) (cong (σ′ ⇒_) q))
              ; _ _ → nothing }
  ι  ==T ι  = just refl
  o  ==T o  = just refl
  _  ==T _  = nothing

  _==V_ : {σ τ : Types} → 𝒱 σ → 𝒱 τ → Maybe (σ ≡ τ)
  var i σ ==V var i′ τ =
    if i == i′ then σ ==T τ else nothing
```

4 PCF.Terms

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}

module PCF.Terms where

open import PCF.Domain-Notation
  using (⟪_⟫; _→ᶜ_; _→ˢ_)
open import PCF.Types
  using (Types; _⇒_; σ; 𝒟)
open import PCF.Constants
  using (ℒ; 𝒜⟦_⟧; c)
open import PCF.Variables
  using (𝒱; Env; _⟦_⟧)
open import PCF.Environments
  using (_[_/_])

-- Syntax

data Terms : Types → Set where
  V     : {σ : Types} → 𝒱 σ → Terms σ                          -- variables
  L     : {σ : Types} → ℒ σ → Terms σ                          -- constants
  _␣_   : {σ τ : Types} → Terms (σ ⇒ τ) → Terms σ → Terms τ   -- application
  ƛ_␣_  : {σ τ : Types} → 𝒱 σ → Terms τ → Terms (σ ⇒ τ)       -- λ-abstraction

variable M N : Terms σ
infixl 20 _␣_

-- Semantics

𝒜′⟦_⟧ : Terms σ → ⟪ Env →ˢ 𝒟 σ ⟫

𝒜′⟦ V α     ⟧ ρ = ρ ⟦ α ⟧
𝒜′⟦ L c     ⟧ ρ = 𝒜⟦ c ⟧
𝒜′⟦ M ␣ N   ⟧ ρ = 𝒜′⟦ M ⟧ ρ (𝒜′⟦ N ⟧ ρ)
𝒜′⟦ ƛ α ␣ M ⟧ ρ = λ x → 𝒜′⟦ M ⟧ (ρ [ x / α ])
```

5 PCF.Types

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}

module PCF.Types where

open import Data.Bool.Base
  using (Bool)
open import Agda.Builtin.Nat
  using (Nat)

open import PCF.Domain-Notation
  using (Domain; ⟪_⟫; _→ᶜ_; _+⊥)

-- Syntax

data Types : Set where
  ι       : Types                   -- natural numbers
  o       : Types                   -- Boolean truthvalues
  _⇒_     : Types → Types → Types   -- functions

variable σ τ : Types

infixr 1 _⇒_

-- Semantics 𝒟

𝒟 : Types → Domain

𝒟 ι       = Nat +⊥
𝒟 o       = Bool +⊥
𝒟 (σ ⇒ τ) = 𝒟 σ →ᶜ 𝒟 τ

variable x y z : ⟪ 𝒟 σ ⟫
```

6 PCF.Variables

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}

module PCF.Variables where

open import Agda.Builtin.Nat
  using (Nat)

open import PCF.Domain-Notation
  using (⟪_⟫; _→ᶜ_; _→ˢ_)
open import PCF.Types
  using (Types; σ; 𝒟)

-- Syntax

data 𝒱 : Types → Set where
  var : Nat → (σ : Types) → 𝒱 σ

variable α : 𝒱 σ

-- Environments

Env = {σ : Types} → 𝒱 σ → ⟪ 𝒟 σ ⟫

variable ρ : Env

-- Semantics

_⟦_⟧ : ⟪ Env →ˢ 𝒱 σ →ˢ 𝒟 σ ⟫

ρ ⟦ α ⟧ = ρ α
```

7 PCF.index

```
{-# OPTIONS --rewriting --confluence-check --lossy-unification #-}

module PCF.index where

import PCF.Domain-Notation
import PCF.Types
import PCF.Constants
import PCF.Variables
import PCF.Environments
import PCF.Terms
-- import PCF.Checks
```